

# Q + Bert


**Die VC 20-Version des bekannten Spielhallenhits. Nur besteht diese Variante aus drei verschiedenen Bildern.**

Ziel des Spieles ist es, genau wie im bekannten Original, sämtliche Felder der Pyramide umzufärben. Das erreicht man, indem man einfach auf die noch unbeschränkten Felder springt. Damit es nicht zu einfach wird, ist noch eine Schlange mit im Spiel, mit der man nicht in Kontakt kommen darf. Wenn man es geschafft hat, alle Felder umzufärben, ohne die Schlange zu berühren, gelangt man in die nächste Pyramide. Nachdem auch die dritte Pyramide bewältigt ist, wechseln die Farben, und man fängt wieder mit der ersten Pyramide an, diesmal jedoch mit einem höheren Schwierigkeitsgrad.

Q + Bert wird mit folgenden Tasten gesteuert:

- A — für einen Sprung nach links oben
- F — für einen Sprung nach rechts oben
- C — für einen Sprung nach links unten
- Z — für einen Sprung nach rechts unten

Diese Steuerung ist etwas gewöhnungsbedürftig, eine Routine zur Abfrage des Joysticks hätte den Programmablauf jedoch stark verlangsamt. Außerdem können die Tasten in den Zeilen 320 bis 350 leicht geändert werden.



Das Spielfeld im ersten Bild

Bei den ersten beiden Pyramiden muß keine besondere Strategie angewendet werden. Zu beachten ist lediglich, daß man sich nur an der Schlange vorbeibewegen kann, nachdem sie aufgeblinkt hat. Weiter ist es hilfreich zu wissen, daß die Schlange sich nur dann bewegt, wenn sowohl deren X-Position als auch deren Y-Position nicht mit der eigenen X-Position beziehungsweise Y-Position übereinstimmt. In der dritten Pyramide kann man sich nicht an der Schlange vorbeimogeln. Aus diesem Grund muß man zuerst alle Felder umfärben, auf die die Schlange nicht gelangen kann und sie dann auf die umgefärbten Felder locken. Danach können die restlichen Felder umgefärbt werden.

Falls das Spiel so zu leicht erscheint, kann in Zeile 580 der Schwierigkeitsgrad erhöht werden, indem »S\*2« durch »S\*3« ersetzt wird.

(Oliver Pabst/ev)

Zeile	Funktion
10	Herabsetzen des für Basic verfügbaren Speicherplatzes
25-45	Erstellung des neuen Zeichensatzes
46-60	Festlegen der Variablen und Setzen der Farben
75	Aufruf zum Drucken der Pyramide
80-90	Initialisieren des Farb-RAMs (Modus für mehrfarbige Zeichen)
300-350	Tastaturabfrage
360-370	Tastaturpuffer entleeren / Schrittgeräusch
410-415	Fehlende Teile der Pyramide an die alte Position POKEn
420-421	Zeichen der Pyramide an der neuen Position ermitteln
424-430	Q+BERT am Rand der Pyramide?
440-445	Q+BERT POKEn
500-505	Feld schon umgefärbt?
520	Punkttestand ausdrucken
530-560	Gesamte Pyramide umgefärbt?
580	Gegner bewegen? (RD kleiner 21-S*2)
600-630	Neue Position des Gegners errechnen
660-666	Fehlende Teile der Pyramide an die alte Position POKEn
670-675	Zeichen der Pyramide an der neuen Position ermitteln
676-677	Gegner am Rand der Pyramide?
678-680	Gegner POKEn
1000-1500	Erste Pyramide drucken
2000-2050	Feld umfärben / Punkttestand erhöhen)
3000-3200	Zweite Pyramide drucken
4000-4200	Dritte Pyramide POKEn
5000-5310	Unterprogramme, die testen, ob die Bewegung des Gegners möglich ist, oder er von der Pyramide springt
8000-8050	Unterprogramm zum Abziehen der Punkte
10000-10200	Daten für die Umprogrammierung des Zeichensatzes
20000-20060	Fall von der Pyramide
20080-20100	Wackeln des Bildschirms
20200-20260	Abfrage, ob noch ein Spiel und Ausgabe der erreichten Punkte
30000-30020	Gegner fängt Q+BERT

## Der Programmaufbau von »Q+Bert«

U	— Entscheidet, welche Pyramide gedruckt wird
RD	— Entscheidet, ob Gegner bewegt wird
S	— Farbe und Schwierigkeitsgrad
X	— Eigene X-Position
Y	— Eigene Y-Position
D	— X-Position des Gegners
DD	— Y-Position des Gegners
PU	— Anzahl der umgefärbten Felder
D1	— Zeichen der Pyramide unter dem Gegner
D2	— Zeichen der Pyramide unter dem Gegner
C1	— Zeichen der Pyramide unter Q+Bert
C2	— Zeichen der Pyramide unter Q+Bert
XX	— Eigene alte X-Position
YY	— Eigene alte Y-Position
S1	— Zeichen Q+BERT (links oben)
S2	— Zeichen Q+BERT (rechts oben)
AD	— Alte X-Position des Gegners
SD	— Alte Y-Position des Gegners
O1	— Zeichen Gegner (links oben)
O2	— Zeichen Gegner (rechts oben)
C3	— Zeichen der Pyramide unter Q+BERT (nur beim Fall)
C4	— Zeichen der Pyramide unter Q+BERT (nur beim Fall)
F1	— Hilfsfarbe
F2	— Farbe, in die umgefärbt wird

## Variablen-tabelle

```

0 REM ===== <253>
1 REM Q+BERT <057>
2 REM ===== <255>
3 REM OLIVER FABST <221>
4 REM LUETTGENBERG 2 <077>
5 REM 3320 SALZGITTER 1 <150>
6 REM TEL. 05341/54845 <222>
7 REM <150>
10 POKE 52,24:POKE 56,24:CLR <062>
20 POKE 649,1 <171>
21 PRINT "{CLR}" <133>
25 FOR I=6400 TO 6408:POKE I,0:NEXT I <080>
30 FOR I=0 TO 175 <011>
40 READ A:POKE 6152+I,A:NEXT I <244>
45 FOR I=376 TO 456:POKE 6152+I,PEEK(32776+I)
:NEXT:P=0 <129>
46 POKE 36869,254:U=1:TI$="000000":RD=15:S=3
<121>
50 POKE 36879,9 <063>
52 F1(3)=2:F2(3)=12:F1(4)=6:F2(4)=15:F1(5)=8
:F2(5)=14 <059>
55 F1(6)=2:F2(6)=12 <103>
60 POKE 36878,F1(5)*16 OR 15 <083>
65 IF P=0 THEN 72 <014>
70 POKE 198,0:GOSUB 8000 <106>
72 PRINT "{CLR}":RD=15 <160>
75 ON U GOSUB 1000,3000,4000 <094>
80 FOR I=38421 TO 38905:POKE I,11 <035>
90 NEXT I <037>
210 PU=0:D1=8:D2=7 <139>
220 TI$="000000":C1=8:C2=7 <227>
230 POKE 198,0:IF U=3 THEN C1=1:C2=2 <011>
240 A$="":GOTO 310 <162>
300 GET A$:IF A$="" THEN 500 <143>
310 XX=X:YY=Y <232>
320 IF A$="Z" THEN X=X+1:Y=Y+66 <025>
330 IF A$="A" THEN X=X-1:Y=Y-66 <012>
340 IF A$="C" THEN X=X-1:Y=Y+66 <023>
350 IF A$="F" THEN X=X+1:Y=Y-66 <036>
360 POKE 198,0:POKE 36874,128 <162>
370 POKE 36875,138:POKE 36874,0:POKE 36875,0
<082>
410 POKE 7680+XX+YY,C1:POKE 7680+XX+YY+1,C2
<053>
415 POKE 7680+XX+YY+22,3:POKE 7680+XX+YY+23,4
<250>
420 C1=PEEK(7680+X+Y) <184>
421 C2=PEEK(7680+X+Y+1) <149>
424 S1=15:S2=16 <029>
425 IF C1=1 THEN S1=11 <203>
426 IF C2=2 THEN S2=12 <208>
430 IF C1<>1 AND C1<>8 THEN 20000 <155>
440 POKE 7680+X+Y,S1:POKE 7680+X+Y+1,S2 <017>
445 POKE 7680+X+Y+22,13:POKE 7680+X+Y+23,14
<024>
500 A=PEEK(38400+X+Y) AND 7 <229>
505 IF A=3 THEN GOSUB 2000 <161>
520 PRINT "{HOME,WHITE}";INT(P*25);"{HOME}
"TAB(17);RIGHT$(TI$,4) <223>
530 IF PU=21 AND U=1 THEN U=2:GOTO 70 <074>
550 IF U=2 AND PU=37 THEN U=3:GOTO 70 <103>
560 IF U=3 AND PU=28 THEN U=1:S=S+1:GOTO 60
<220>
570 RD=RD+1 <245>
580 IF RD<21-S*2 THEN 800 <152>
590 AD=D:SD=DD <214>
600 IF X>D AND Y>DD THEN D=D+1:DD=DD+66
:GOSUB 5000 <205>
610 IF X>D AND Y<DD THEN D=D+1:DD=DD-66
:GOSUB 5100 <219>
620 IF X<D AND Y<DD THEN D=D-1:DD=DD-66
:GOSUB 5200 <233>
630 IF X<D AND Y>DD THEN D=D-1:DD=DD+66
:GOSUB 5300 <241>
660 POKE 7680+AD+SD,D1:POKE 7680+AD+SD+1,D2
<166>
666 POKE 7680+AD+SD+22,3:POKE 7680+AD+SD+23,4
<106>
667 IF Z=S1 THEN 30000 <082>
670 D1=PEEK(7680+D+DD) <207>
675 D2=PEEK(7680+D+DD+1) <176>
676 O1=17:O2=18:IF D1=1 THEN O1=21 <111>
677 IF D2=2 THEN O2=22 <202>
678 POKE 7680+D+DD,O1:POKE 7680+D+DD+1,O2 <046>
680 POKE 7680+D+DD+22,19:POKE 7680+D+DD+23,20
<061>
690 RD=0 <044>
800 GOTO 300 <031>
1000 X=10:Y=9*22:D=10:DD=15*22 <095>
1010 PRINT:PRINT <097>
1020 PRINT "{10SPACE}AB" <095>
1030 PRINT "{10SPACE}CD" <110>
1040 PRINT "{10SPACE}EF" <124>
1050 PRINT "{9SPACE}AGHB" <013>
1060 PRINT "{9SPACE}CDCD" <019>
1070 PRINT "{9SPACE}EFEF" <037>
1080 PRINT "{8SPACE}AGHGH" <186>
1090 PRINT "{8SPACE}CDCDCD" <184>
1100 PRINT "{8SPACE}EFEFEF" <206>
1110 PRINT "{7SPACE}AGHGHGH" <103>
1120 PRINT "{7SPACE}CDCDCDCD" <093>
1130 PRINT "{7SPACE}EFEFEFEF" <119>
1140 PRINT "{6SPACE}AGHGHGHGH" <020>
1150 PRINT "{6SPACE}CDCDCDCDCD" <002>
1160 PRINT "{6SPACE}EFEFEFEFEF" <032>
1170 PRINT "{5SPACE}AGHGHGHGHGH" <193>
1180 PRINT "{5SPACE}CDCDCDCDCDCD" <167>
1190 PRINT "{5SPACE}EFEFEFEFEFEF" <201>
1200 PRINT "{5SPACE}IJIJIJIJIJIJ" <003>
1500 RETURN <111>
2000 POKE 36875,180 <064>
2005 PU=PU+1:P=P+1 <026>
2010 POKE 38400+X+Y,F2(S) <196>
2020 POKE 38400+X+1+Y,F2(S) <169>
2030 POKE 38400+X+Y+22,F2(S) <230>
2040 POKE 38400+X+23+Y,F2(S) <241>
2045 POKE 36875,0 <004>
2050 RETURN <152>
3000 X=10:Y=6*22:D=10:DD=18*22 <055>
3005 PRINT:PRINT <052>
3010 PRINT "{9SPACE}ABAB" <142>
3015 PRINT "{9SPACE}CDCD" <155>
3020 PRINT "{9SPACE}EFEF" <168>
3030 PRINT "{2SPACE}ABAB {2SPACE}AGHGH {2SPACE}
ABAB" <073>
3040 PRINT "{2SPACE}CDCD {2SPACE}CDCDCD {2SPACE}
CDCD" <087>
3050 PRINT "{2SPACE}EFEF {2SPACE}EFEFEF {2SPACE}
EFEF" <125>
3060 PRINT " AGHGHBAGHGHGHGHGH" <044>
3070 PRINT " CDCDCDCDCDCDCDCDCD" <010>
3080 PRINT " EFEFEFEFEFEFEFEFEF" <061>
3090 PRINT " IJIHGHGHGHGHGHGHIJ" <123>
3100 PRINT "{4SPACE}CDCDCDCDCDCDCD" <148>
3110 PRINT "{4SPACE}EFEFEFEFEFEFEF" <186>
3120 PRINT "{4SPACE}IHGHGHGHGHGHGJ" <228>
3130 PRINT "{5SPACE}CDCDCDCDCDCD" <043>
3140 PRINT "{5SPACE}EFEFEFEFEFEF" <077>
3150 PRINT "{5SPACE}IHGHGHGHGHGHGJ" <115>
3160 PRINT "{6SPACE}CDCDCDCDCD" <194>
3170 PRINT "{6SPACE}EFEFEFEFEF" <224>
3180 PRINT "{6SPACE}IJIJIJIJIJ" <018>
3200 RETURN <026>
4000 PRINT:PRINT:X=10:Y=12*22:D=10:DD=6*22 <195>
4005 PRINT "{5SPACE}ABABABABAB" <129>
4010 PRINT "{5SPACE}CDCDCDCDCDCD" <158>

```

Listing »Q+Bert«

```

4020 PRINT "{5SPACE}EFEFEFEFEFEF          <192>
4030 PRINT "{4SPACE}AGHGHGHGHGHGHGB      <101>
4040 PRINT "{4SPACE}CDCDCDCDCDCDCDCD      <067>
4050 PRINT "{4SPACE}EFEFEFEFEFEFEFEF     <105>
4060 PRINT "{3SPACE}AGJIJIJIJIJIJIJH      <042>
4070 PRINT "{3SPACE}CD{12SPACE}CD         <190>
4080 PRINT "{3SPACE}EF{12SPACE}EF          <208>
4090 PRINT "{3SPACE}IHBABABABABABAGJ      <248>
4100 PRINT "{4SPACE}CDCDCDCDCDCDCDCD      <128>
4110 PRINT "{4SPACE}EFEFEFEFEFEFEFEF     <166>
4120 PRINT "{4SPACE}IHGHGHGHGHGHGHGJ      <208>
4130 PRINT "{5SPACE}CDCDCDCDCDCDCD        <023>
4140 PRINT "{5SPACE}EFEFEFEFEFEFEFEF     <057>
4150 PRINT "{5SPACE}IJIJIJIJIJIJIJIJ      <115>
4200 RETURN                                <006>
5000 Z=PEEK(7680+D+DD):IF Z>8 AND Z<>S1 THEN D=
    D-1:DD=DD-66                            <120>
5010 RETURN                                <051>
5100 Z=PEEK(7680+D+DD):IF Z>8 AND Z<>S1 THEN D=
    D-1:DD=DD+66                            <219>
5110 RETURN                                <151>
5200 Z=PEEK(7680+D+DD):IF Z>8 AND Z<>S1 THEN D=
    D+1:DD=DD+66                            <063>
5210 RETURN                                <252>
5300 Z=PEEK(7680+D+DD):IF Z>8 AND Z<>S1 THEN D=
    D+1:DD=DD-66                            <164>
5310 RETURN                                <096>
9000 PRINT "{WHITE,HOME,7SPACE}"          <084>
9005 FOR I=1 TO 2*VAL(TI#)                  <106>
9010 P=P-0.2                                <246>
9020 PRINT "{HOME}";INT(P*25)              <007>
9030 NEXT I                                  <072>
9040 FOR I=1 TO 1800:NEXT                  <093>
9050 RETURN                                  <031>
10000 DATA 2,2,10,10,42,170,170          <016>
10010 DATA 128,128,160,160,168,168,170,170 <202>
10020 DATA 170,170,106,106,90,90,86,86    <014>
10030 DATA 170,170,171,171,175,175,191,191 <222>
10040 DATA 85,85,85,85,85,85,85,85        <126>
10050 DATA 255,255,255,255,255,255,255,255 <000>
10060 DATA 149,149,165,165,169,169,170,170 <014>
10070 DATA 254,254,250,250,234,234,170,170 <250>
10080 DATA 85,85,21,21,5,5,1,1           <170>
10090 DATA 255,255,252,252,240,240,192,192 <022>
10100 DATA 2,2,15,15,47,44,188,191        <145>
10110 DATA 128,128,224,224,248,248,254,254 <052>
10120 DATA 175,175,111,111,90,90,86,86    <116>
10130 DATA 238,234,235,251,175,175,191,191 <072>
10135 DATA 254,254,251,251,239,236,188,191 <080>
10140 DATA 149,149,229,229,233,233,254,254 <086>
10150 DATA 254,254,240,240,196,192,160,168 <088>
10160 DATA 149,149,165,165,169,41,42,42   <211>
10170 DATA 168,170,106,96,82,80,86,86     <131>
10180 DATA 10,10,11,43,175,47,191,191     <107>
10190 DATA 2,2,0,0,4,0,160,168            <010>
10200 DATA 128,128,160,160,168,40,42,42   <233>
20000 FOR I=1 TO 21-Y/22                   <023>
20010 C1=PEEK(7680+X+Y+I*22)
      :C2=PEEK(7680+X+Y+1+I*22)            <186>
20015 C3=PEEK(7680+X+Y+22+I*22)
      :C4=PEEK(7680+X+Y+23+I*22)          <005>
20020 POKE 7680+X+Y+I*22,11                <132>
20021 POKE 7680+X+Y+1+I*22,12             <097>
20022 POKE 7680+X+Y+22+I*22,13           <150>
20023 POKE 7680+X+Y+23+I*22,14           <153>
20040 POKE 7680+X+Y+I*22,C1               <170>
20041 POKE 7680+X+Y+1+I*22,C2            <135>
20042 POKE 7680+X+Y+22+I*22,C3           <188>
20043 POKE 7680+X+Y+23+I*22,C4           <191>
20060 NEXT I                                <117>
20080 FOR I=1 TO 32                         <116>
20090 POKE 36865,38:POKE 36865,39:POKE 36865,40
      :POKE 36865,39                        <089>
20100 NEXT                                  <084>
20200 POKE 36879,27:PRINT "{CLR,BLUE}"     <060>
20210 POKE 36869,240                         <169>
20215 PRINT "{8DOWN,SPACE}IHRE PUNKTE: ";INT(P*25)
      <135>
20220 PRINT "{2DOWN,SPACE}NOCH EIN SPIEL (J/N)"
      <226>
20230 GET A$:IF A#=""THEN 20230              <025>
20240 IF A#="N"THEN END                       <186>
20250 IF A#="J"THEN CLR:GOTO 46              <009>
20260 GOTO 20230                              <243>
30000 POKE 7680+D+DD,01:POKE 7680+D+DD+1,02
      <043>
30010 POKE 7680+D+DD+22,19:POKE 7680+D+DD+23,20
      <066>
30020 GOTO 20000                              <052>

```

## Listing »Q+Bert« (Schluß)

